# $Id$
# Maintainer: Jan de Groot <jgc@archlinux.org>
# Contributor: Arjan Timmerman <arjan@archlinux.org>

pkgname=libxklavier
pkgver=5.4
pkgrel=1
pkgdesc="High-level API for X Keyboard Extension"
arch=(i686 x86_64)
license=('LGPL')
depends=('glib2' 'libxkbfile' 'libxml2' 'xkeyboard-config' 'libxi' 'iso-codes')
makedepends=('gobject-introspection' 'intltool' 'gtk-doc' 'libxfixes')
options=('!emptydirs')
url="http://gswitchit.sourceforge.net"
source=("http://cgit.freedesktop.org/$pkgname/snapshot/$pkgname-$pkgver.zip")
sha256sums=('d5d2c135e3227142f5b455b9ffdf3e05be4b1e85a6c2882f8e905cc30c362d3e')

build() {
  cd "$pkgname-$pkgver"
  ./autogen.sh --prefix=/usr --with-xkb-bin-base=/usr/bin \
      --with-xkb-base=/usr/share/X11/xkb --disable-static --enable-gtk-doc
  make
}

package() {
  cd "$pkgname-$pkgver"
  make DESTDIR="$pkgdir" install
}
